Take the 2-minute tour ×

If I define a constant in a Perl module, how do I use that constant in my main program? (Or how do I call that constant in the main program?)

share|improve this question
Best to avoid the "constant" package. earino.wordpress.com/2013/02/27/… Alternatives here: neilb.org/reviews/constants.html –  nslntmnx Mar 28 '13 at 0:37

6 Answers 6

Constants can be exported just like other package symbols. Using the standard Exporter module, you can export constants from a package like this:

package Foo;
use strict;
use warnings;

use base 'Exporter';

use constant CONST => 42;

our @EXPORT_OK = ('CONST');


Then, in a client script (or other module)

use Foo 'CONST';
print CONST;

You can use the %EXPORT_TAGS hash (see the Exporter documentation) to define groups of constants that can be exported with a single import argument.

Update: Here's an example of how to use the %EXPORT_TAGS feature if you have multiple constants.

use constant LARRY => 42;
use constant CURLY => 43;
use constant MOE   => 44;

our @EXPORT_OK = ('LARRY', 'CURLY', 'MOE');
our %EXPORT_TAGS = ( stooges => [ 'LARRY', 'CURLY', 'MOE' ] );

Then you can say

use Foo ':stooges';
print "$_\n" for LARRY, CURLY, MOE;
share|improve this answer
use Exporter 'import'; is better, but will require upgrading Exporter on perl's earlier than 5.8.3. –  Alexandr Ciornii Feb 14 '09 at 0:33
please explain why is using 'import' better? –  Matthew Watson Jun 12 '09 at 5:29
@nsIntmnx Neither of those links worked for me. –  Technophile Apr 21 '14 at 21:18

Constants are just subs with empty prototype, so they can be exported like any other sub.

# file Foo.pm
package Foo;
use constant BAR => 123;
use Exporter qw(import);
our @EXPORT_OK = qw(BAR);

# file main.pl:
use Foo qw(BAR);
print BAR;
share|improve this answer

To expand on the earlier answers, since constants are really just subs, you can also call them directly:

use Foo;
print Foo::BAR;
share|improve this answer
Don't refer to answers as being above, because if your answer gets voted up enough, it may end up above them. –  Brad Gilbert Oct 14 '08 at 17:43

You might want to consider using Readonly instead of constant.

share|improve this answer
Why? What's wrong with constant? –  moritz Oct 11 '08 at 11:05
You can't do this with it: print "constant is $constant"; or this: print $hash{constant}; –  Leon Timmermans Oct 11 '08 at 11:35
Since 'use constant' creates a subroutine, there are unexpected issues => no string interpolation, difficulty in using as a hash key. See the docs for Readonly [ metacpan.org/module/Readonly#COMPARISON-WITH-use-constant ] . Readonly use the perl internals to directly mark a variable as read-only. Readonly is slow (unless using Readonly::XS) and unmaintained. See also Data::Lock and Const::Fast for different implementations. –  spazm Nov 11 '11 at 22:31
Actually you can interpolate constant in a string. Just use the same syntax you would use with any other function: use constant COLOR => 'red'; print "The color is @{[COLOR]}!"; –  Tero Niemi Mar 30 '13 at 22:54
package Foo;
use Readonly;
Readonly my  $C1 => 'const1';
Readonly our $C2 => 'const2';
sub get_c1 { return $C1 }

perl -MFoo -e 'print "$_\n" for Foo->get_c1, $Foo::C2'
share|improve this answer

To add to the bag of tricks, since a constant is just a subroutine you can even call it as a class method.

package Foo;
use constant PI => 3.14;

print Foo->PI;

If you have lots of constants it's a nice way to get at the occasional one without having to export them all. However, unlike Foo::PI or exporting PI, Perl will not compile out Foo->PI so you incur the cost of a method call (which probably doesn't matter).

share|improve this answer

Your Answer


